
Ściąga DAX · zajęcia 4
Kod z pliku INWD_dax_rozwiazanie.pbip (tabela _Miary). W swoim projekcie podstaw własne tabele i kolumny.

1 · Składnia i nawyki

• 'Tabela'[Kolumna] — zawsze z nazwą tabeli; [Miara] — bez niej

• VAR _nazwa = … RETURN … — wynik pośredni liczony raz

• DIVIDE(a, b) zamiast a / b — pusto zamiast błędu przy zerze

• format przy mierze: #,0 dla złotych, 0,0% dla udziałów

• każda miara w tabeli _Miary, z opisem i w folderze

Marża % =

VAR _sprzedaz = [Sprzedaż PLN]

VAR _koszt = [Koszt PLN]

RETURN

DIVIDE(_sprzedaz - _koszt, _sprzedaz)

2 · Dwa konteksty

• kontekst filtra: filtry komórki (wiersz, kolumna, fragmentatory, 
filtry strony) płyną relacjami do faktów

• kontekst wiersza: bieżący wiersz w kolumnie obliczeniowej 
albo w iteratorze; sam nie filtruje

• przejście kontekstu: miara albo CALCULATE w iteratorze 
zamienia wiersz na filtr

• wiersz sumy to osobny kontekst — sprawdzaj go zawsze

3 · Kontrola miary w pięciu krokach

• przeczytaj na głos: co filtruje, co iteruje, co zwraca

• przewidź wynik jednej komórki

• sprawdź zapytaniem EVALUATE na małym wycinku

• sprawdź sumę, puste dane i zero w mianowniku

• poruszaj fragmentatorem i porównaj wynik z manifestem

4 · Typowe błędy w DAX-ie z czatu AI

• 'Sales'[Amount] zamiast twoich nazw — podaj AI tabele 
i kolumny swojego modelu

• SUM w iteratorze liczy cały kontekst — użyj miary albo 
CALCULATE

• udział nie reaguje na fragmentator — ALLSELECTED zamiast 
REMOVEFILTERS

• FILTER(ALL('Tabela'), …) — wystarczy prosty filtr kolumny, 
czasem z KEEPFILTERS

• a / b — DIVIDE(a, b)

• COUNTIFS, IFS, VLOOKUP — w DAX-ie ich nie ma: 
CALCULATE(COUNTROWS(…)), SWITCH(TRUE()), RELATED

5 · CALCULATE: trzy ruchy na filtrach

Dodaj filtr
Zastępuje filtr tej kolumny, reszta zostaje.

Sprzedaż online PLN =

CALCULATE([Sprzedaż PLN], 'Kanał'[Typ kanału] = "Online")

Usuń filtr: udział w całości
Mianownik bez filtrów produktu, także z fragmentatora.

Sprzedaż wszystkich produktów PLN =

CALCULATE([Sprzedaż PLN], REMOVEFILTERS('Produkt'))

Udział w całości % =

DIVIDE([Sprzedaż PLN], [Sprzedaż wszystkich produktów PLN])

Zostaw wybór użytkownika: udział w wybranych
Zdejmuje filtr wiersza, zostawia fragmentator.

Udział w wybranych % =

DIVIDE([Sprzedaż PLN],

CALCULATE([Sprzedaż PLN], ALLSELECTED('Produkt')))

KEEPFILTERS: dołóż zamiast podmieniać
Wartość tylko tam, gdzie filtry się przecinają.

Komputery KEEPFILTERS PLN =

CALCULATE([Sprzedaż PLN],

KEEPFILTERS('Produkt'[Kategoria] = "Komputery"))

6 · Iteratory i ranking

SUMX: wiersz po wierszu, potem suma
Sprzedaż brutto PLN =

SUMX('Sprzedaż',

'Sprzedaż'[Ilość] * 'Sprzedaż'[Cena jednostkowa])

AVERAGEX: średnia z dni (w środku miara!)
Średnia dzienna sprzedaż PLN =

AVERAGEX(VALUES('Kalendarz'[Data]), [Sprzedaż PLN])

RANKX + ISINSCOPE: pusto w wierszu sumy
Ranking podkategorii =

IF(

ISINSCOPE('Produkt'[Podkategoria]),

RANKX(ALLSELECTED('Produkt'[Podkategoria]),

[Sprzedaż PLN])

)

7 · Widok zapytań DAX

Test na małym wycinku i spis miar
EVALUATE

SUMMARIZECOLUMNS(

'Kalendarz'[Rok],

"Marża %", [Marża %],

"Udział rabatów %", [Udział rabatów %]

)

EVALUATE INFO.VIEW.MEASURES()

Ściąga DAX · pięć funkcji i nowości 2026
SELECTEDVALUE, SWITCH(TRUE()), TOPN + CONCATENATEX, UNICHAR + FORMAT, funkcje okna, UDF, obliczenia wizualne.

8 · Pięć ciekawych funkcji

SELECTEDVALUE: jeden rok z filtra
Sprzedaż rok wcześniej PLN =

VAR _rok = SELECTEDVALUE('Kalendarz'[Rok])

RETURN

CALCULATE([Sprzedaż PLN], 'Kalendarz'[Rok] = _rok - 1)

Zmiana r/r z VAR i DIVIDE
Zmiana r/r % =

VAR _teraz = [Sprzedaż PLN]

VAR _wczesniej = [Sprzedaż rok wcześniej PLN]

RETURN

DIVIDE(_teraz - _wczesniej, _wczesniej)

Tytuł z miary (Tytuł → fx → Wartość pola)
Tytuł dynamiczny =

VAR _okres =

SELECTEDVALUE('Kalendarz'[Rok], "w wybranych latach")

VAR _zmiana = [Zmiana r/r %]

RETURN

"Sprzedaż " & _okres & ": "

& FORMAT([Sprzedaż PLN] / 1000000, "#,0.0")

& " mln zł"

& IF(NOT ISBLANK(_zmiana),

" (" & FORMAT(_zmiana, "+0.0%;-0.0%") & " r/r)")

SWITCH(TRUE()): pierwszy prawdziwy warunek
Segment podkategorii =

VAR _udzial = [Udział w całości %]

RETURN

SWITCH(

TRUE(),

NOT HASONEVALUE('Produkt'[Podkategoria])

|| ISBLANK(_udzial), BLANK(),

_udzial >= 0.10, "Lider",

_udzial >= 0.03, "Środek",

"Ogon"

)

9 · Dla ambitnych: funkcje okna

OFFSET: wiersz wcześniej
Sprzedaż poprzedni kwartał PLN =

CALCULATE([Sprzedaż PLN],

OFFSET(-1,

ALLSELECTED('Kalendarz'[Rok i kwartał],

'Kalendarz'[Klucz kwartału]),

ORDERBY('Kalendarz'[Klucz kwartału], ASC)))

WINDOW: okno trzech kwartałów
Średnia z 3 kwartałów PLN =

AVERAGEX(

WINDOW(-2, REL, 0, REL,

ALLSELECTED('Kalendarz'[Rok i kwartał],

'Kalendarz'[Klucz kwartału]),

ORDERBY('Kalendarz'[Klucz kwartału], ASC)),

[Sprzedaż PLN])

8 · Pięć ciekawych funkcji (ciąg dalszy)

TOPN + CONCATENATEX: lista jako tekst
Top 3 podkategorie =

CONCATENATEX(

TOPN(3, VALUES('Produkt'[Podkategoria]),

[Sprzedaż PLN]),

'Produkt'[Podkategoria],

", ",

[Sprzedaż PLN], DESC

)

UNICHAR + FORMAT: ▲ +15,9% / ▼ -8,0%
Trend r/r =

VAR _zmiana = [Zmiana r/r %]

RETURN

IF(

NOT ISBLANK(_zmiana),

IF(_zmiana >= 0, UNICHAR(9650), UNICHAR(9660))

& " " & FORMAT(_zmiana, "+0.0%;-0.0%")

)

Kolor czcionki z miary (Elementy komórki → fx)
Kolor trendu =

VAR _zmiana = [Zmiana r/r %]

RETURN

SWITCH(TRUE(), ISBLANK(_zmiana), BLANK(),

_zmiana < 0, "#C0392B", "#18480C")

10 · Nowości 2026: UDF i obliczenia wizualne

Funkcja użytkownika w widoku zapytań DAX
Ogólnodostępne od czerwca 2026. Zapis w modelu wymaga poziomu 
zgodności 1702 lub wyższego.

DEFINE

FUNCTION ZmianaProcent =

(teraz : NUMERIC, wczesniej : NUMERIC) =>

DIVIDE(teraz - wczesniej, wczesniej)

EVALUATE

SUMMARIZECOLUMNS(

'Kalendarz'[Rok],

"Zmiana r/r",

ZmianaProcent([Sprzedaż PLN],

[Sprzedaż rok wcześniej PLN])

)

Obliczenia wizualne (Nowe obliczenie wizualne)
Liczą tylko na polach z wykresu; nie użyjesz ich w innej wizualizacji ani 
mierze.

Średnia krocząca = MOVINGAVERAGE([Sprzedaż PLN], 3)

Narastająco = RUNNINGSUM([Sprzedaż PLN])

Zmiana k/k = [Sprzedaż PLN] - PREVIOUS([Sprzedaż PLN])

